Building a stock-flow consistent model in practice

Yannis Dafermos (University of the West of England)
Maria Nikolaidi (University of Greenwich)

PhD lecture series in advanced macroeconomics, University of Greenwich, 27 May 2015

Model 1

Model equations:

Households and production sector:
Consumption expenditures: $C = c_1 Y_{-1} + c_2 r M_{-1}$
Money or government debt (identity): $M = M_{-1} + Y + rM_{-1} - C$
Income (identity): $Y = C + G$

Government:
Government expenditures: $G = G_{-1}(1+gg)$
Money or government debt (redundant identity): $M_{red} = M_{-1} + G + rM_{-1}$

'**Step 1:** Open EViews and create a new programme (File->New->Programme)

'**Step 2:** Create a new workfile, which is called 'MODEL 1', identify that this file will run from years 2000 to 3000 and create a time variable which till take values from 2000 to 3000 (when you write the commands, press 'run' and then 'ok')

```
wfcreate(wf=MODEL1) a 2000 3000
genr year=@trend+2000
```

'**Step 3:** Set values for the parameters and the exogenous variables of the model (when you write the commands, press 'run' and then 'ok')

```
'VALUES FOR PARAMETERS & EXOGENOUS VARIABLES
genr c1=0.9
genr c2=0
genr r=0.02
genr gg=0.06
```

'**Step 4:** Set initial values for the endogenous variables of the model (when you write the commands, press 'run' and then 'ok')

```
'INITIAL VALUES FOR ENDOGENOUS VARIABLES
genr Y=100
genr M=60
genr G=10
```

'**Step 5:** Create a new model called 'MODEL 1' and write down the equations of the model (when you write the commands, press 'run' and then 'ok')

```
'MODEL
delete *_model
model MODEL1_model

'Model equations

'Households and production sector
MODEL1_model.append CO=c1*Y(-1)+c2*r*M(-1)
MODEL1_model.append M=M(-1)+Y+r*M(-1)-CO
MODEL1_model.append Y=CO+G

'Government
MODEL1_model.append G=G(-1)*(1+gg)
MODEL1_model.append M_red=M(-1)+G+r*M(-1)

'Additional variables
MODEL1_model.append gy=(Y-Y(-1))/Y(-1)
MODEL1_model.append dratio=M/Y
```

'**Step 6:** Run the model (when you write the commands, press 'run and then 'ok')

```
MODEL1_model.solve
```

'**Step 7:** Group the endogenous variables which are more important for your analysis (when you write the commands, press 'run' and then 'ok')

```
'GROUP DATA
Group data (M_0)(M_red_0)(gy_0)(dratio_0)
Show data
```

'**Step 8:** Create a graph for the debt-to-GDP ratio (when you write the commands, press 'run' and then 'ok')

```
'PLOT DATA
smpl 2000 3000
Plot dratio_0
```

'**Step 9:** Check how your results change when gg is equal to 0.01 instead of 0.06.

Replace `genr gg=0.06` with `genr gg=0.01`

'**Step 10:** Check how your results change when c_2 is equal to 0.5 instead of 0. Keep gg equal to 0.01.

Replace `genr c2=0` with `genr c2=0.8`

Model 2

Model equations:

Households:
Wage income of households: $W = s_w Y$
Capital income of households: $Y_c = DP + BP + r_M M_{-1}$
Consumption expenditures: $C = c_1 W_{-1} + c_2 Y_{c-1} + c_3 M_{-1}$
Deposits (identity): $M = M_{-1} + W + Y_c - C$

Firms:
Income: $Y = C + I$
Total profits of firms (identity): $TP = Y - W - r_L L_{-1}$
Retained profits: $RP = s_f TP$
Distributed profits (identity): $DP = TP - RP$
Investment: $I = g_k K_{-1}$
Capital stock: $K = K_{-1} + I$
Loans (identity): $L = L_{-1} + I - RP$

Banks:
Profits of banks (identity): $BP = r_L L_{-1} - r_M M_{-1}$
Deposits (redundant identity): $M_{red} = L$

'**Step 1:** Open EViews and create a new programme (File->New->Programme)

'**Step 2:** Create a new workfile, which is called 'MODEL 2', identify that this file will run from years 2000 to 2300 and create a time variable which till take values from 2000 to 2300 (when you write the commands, press 'run' and then 'ok')

```
wfcreate(wf=MODEL2) a 2000 2300
genr year=@trend+2000
```

'**Step 3:** Set values for the parameters and the exogenous variables of the model (when you write the commands, press 'run' and then 'ok')

```
genr sw=0.65
genr rm=0.02
genr c1=0.9
genr c2=0.6
genr c3=0.1
genr rl=0.04
genr sf=0.7
genr gk=0.03
```

'**Step 4:** Set initial values for the endogenous variables of the model (when you write the commands, press 'run' and then 'ok')

```
genr M=10
genr Y=10
genr W=sw*Y
genr Yc=3
genr K=20
genr L=M
```

'**Step 5:** Create a new model called 'MODEL 2' and write down the equations of the model (when you write the commands, press 'run' and then 'ok')

```
'MODEL

delete *_model
model MODEL2_model

'Households
MODEL2_model.append W=sw*Y
MODEL2_model.append Yc=DP+BP+rm*M(-1)
MODEL2_model.append CO=c1*W(-1)+c2*Yc(-1)+c3*M(-1)
MODEL2_model.append M=M(-1)+W+Yc-CO

'Firms
MODEL2_model.append Y=CO+I
MODEL2_model.append TP=Y-W-rl*L(-1)
MODEL2_model.append RP=sf*TP
MODEL2_model.append DP=TP-RP
MODEL2_model.append I=gk*K(-1)
MODEL2_model.append K=K(-1)+I
MODEL2_model.append L=L(-1)+I-RP

'Banks
MODEL2_model.append BP=rl*L(-1)-rm*M(-1)
MODEL2_model.append M_red=L

'Additional variables
genr v=0.17
MODEL2_model.append Y_star=v*K
MODEL2_model.append u=Y/Y_star
MODEL2_model.append gy=(Y-Y(-1))/Y(-1)
MODEL2_model.append lev=L/K
```

'**Step 6:** Run the model (when you write the commands, press 'run and then 'ok')

```
'SOLUTION
MODEL2_model.solve
```

'**Step 7:** Group the endogenous variables which are more important for your analysis (when you write the commands, press 'run' and then 'ok')

```
'GROUP DATA
Group data (M_red_0)(M_0)(u_0)(gy_0)(lev_0)
Show data
```

'**Step 8:** Create a graph for the capacity utilisation ratio, the growth rate and the leverage ratio (when you write the commands, press 'run' and then 'ok')

```
'PLOT DATA
smpl 2010 2300
Plot u_0
Plot gy_0
Plot lev_0
```

Step 9: Check how your results change when there is an increase in the wage share from 0.65 to 0.7 in period 2100.

Replace `genr sw=0.65` with `genr sw=0.65*(year<2100)+0.7*(year>=2100)`

Step 10: Check how your results change when there is an increase in the interest rate on loans from 0.04 to 0.06 in period 2100. Use a wage share equal to 0.65.

Replace `genr rl=0.04` with `genr rl=0.04*(year<2100)+0.06*(year>=2100)`

Model 3

Model equations:

Households and production sector:
Output (identity): $Y = C + G$
Disposable income: $YD = Y + rB_{h-1} - T$
Consumption expenditures: $C = c_1 YD_{-1} + c_2 V_{-1}$
Wealth (identity): $V = V_{-1} + YD - C$
Treasury bills held by households: $B_h = (\lambda_0 + \lambda_1 r - \lambda_2 (YD/V_{-1}))V_{-1}$
High-powered money (identity): $HPM = V - B_h$

Government:
Government expenditures: $G = G_{-1}(1+gg)$
Treasury bills (identity): $B = B_{-1} + G + rB_{-1} - T - rB_{cb-1}$
Taxes: $T = \tau YD$

Central bank:
High-powered money (redundant identity): $HPM_{red} = B_{cb}$
Treasury bills held by the central bank (identity): $B_{cb} = B - B_h$

'**Step 1:** Open EViews and create a new programme (File->New->Programme)

'**Step 2:** Create a new workfile, which is called 'MODEL 3', identify that this file will run from years 2000 to 2300 and create a time variable which till take values from 2000 to 2300 (when you write the commands, press 'run' and then 'ok')

```
wfcreate(wf=MODEL3) a 2000 2300
genr year=@trend+2000
```

'**Step 3:** Set values for the parameters and the exogenous variables of the model (when you write the commands, press 'run' and then 'ok'). Use the values reported below.

```
genr r=0.02
genr c1=0.8
genr c2=0.15
genr lambda0=0.7
genr lambda1=0.5
genr lambda2=0.1
genr gg=0.06
genr tau=0.2
```

'**Step 4:** Set initial values for the endogenous variables of the model (when you write the commands, press 'run' and then 'ok')

```
genr Bh=60
genr YD=100
genr B=100
genr Bcb=B-Bh
genr HPM=Bcb
genr V=Bh+HPM
genr G=10
genr Y=120
```

'**Step 5:** Create a new model called 'MODEL 3' and write down the equations of the model (when you write the commands, press 'run' and then 'ok'). In the model include the following additional equations:

```
'MODEL
delete *_model
model MODEL3_model

'Model equations

'Households and production sector
MODEL3_model.append Y=CO+G
MODEL3_model.append YD=Y+r*Bh(-1)-T
MODEL3_model.append CO=c1*YD(-1)+c2*V(-1)
MODEL3_model.append V=V(-1)+YD-CO
MODEL3_model.append Bh=(lambda0+lambda1*r-lambda2*(YD/V(-1)))*V(-1)
MODEL3_model.append HPM=V-Bh

'Government
MODEL3_model.append G=G(-1)*(1+gg)
MODEL3_model.append B=B(-1)+G+r*B(-1)-T-r*Bcb(-1)
MODEL3_model.append T=tau*YD

'Central bank
MODEL3_model.append HPM_red=Bcb
MODEL3_model.append Bcb=B-Bh

'Additional variables
MODEL3_model.append gy=(Y-Y(-1))/Y(-1)
MODEL3_model.append dratio=B/Y
```

'**Step 6:** Run the model (when you write the commands, press 'run and then 'ok')

```
'SOLUTION
MODEL3_model.solve
```

'**Step 7:** Group the variables HPM, HPM_{red}, g_y and $dratio$ (when you write the commands, press 'run' and then 'ok').

```
'GROUP DATA
Group data (HPM_0)(HPM_red_0)(gy_0)(dratio_0)
Show data
```

'**Step 8:** Create the graph for *dratio* (when you write the commands, press 'run' and then 'ok')

```
'PLOT DATA
smpl 2000 2300
Plot dratio_0
```

'**Step 9:** Check how your results change when there is an increase in the tax rate from 0.2 to 0.3 in period 2100.

Replace `genr tau=0.2` with `genr tau=0.2*(year<2100)+0.3*(year>=2100)`

'**Step 10:** Check how your results change when there is an increase in the growth rate of government expenditures from 0.06 to 0.1 in period 2100.

Replace `genr gg=0.06` with `genr gg=0.06*(year<2100)+0.1*(year>=2100)`

'**Step 11:** Check how your results change when there is a decline in λ_0 from 0.7 to 0.5 in period 2100.

Replace `genr lambda0=0.7` with `genr lambda0=0.7*(year<2100)+0.5*(year>=2100)`